Pensieve header: A first implementation of nilpotent integration.

Initialization

```
In[∘]:= SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\HigherRank"];
Once[<< KnotTheory`; << Rot.m];
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
Loading Rot.m from http://drorbn.net/icbs24 to compute rotation numbers.
```

```
In[∘]:= CCF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[ℰ];
CCF[ℰ_] := Factor[ℰ];
CF[ℰ_𝔼] := CF /@ ℰ;
CF[ℰ_List] := CF /@ ℰ;
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := Module[{vs = Cases[ℰ, (x | p)__, ∞] ⋃ {x, p}, ps, c},
   Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) ⧴ CCF[c] (Times @@ vs^ps)] ];
```

Integration

```
In[∘]:= 𝔼 /: 𝔼[A_] 𝔼[B_] := 𝔼[A + B]
```

```
In[∘]:= Unprotect[Integrate];
∫ω_. 𝔼[L_] ⅆ(vs_List) := Module[{n, Q, G, V, s, t, k, a, b},
   n = Length@vs;
   Q = -Table[(∂_{vs⟦a⟧,vs⟦b⟧} L) /. Thread[vs → 0], {a, n}, {b, n}];
   G = Inverse[Q] / 2;
   V = L + vs.Q.vs / 2;
   s = t = V; k = 0;
   While[0 =!= t,
    s += 1/(++k)! (t = CF@Sum[G⟦a, b⟧ ((∂_{vs⟦a⟧,vs⟦b⟧} t) + (∂_{vs⟦a⟧} t) (∂_{vs⟦b⟧} t)), {a, n}, {b, n}])];
   PowerExpand@Factor[ω (Det[Q] (2 π)^n)^(-1/2)] 𝔼[CF@s /. Thread[vs → 0]]
  ];
Protect[Integrate];
```

In[∘]:= $\int \mathbb{E}\left[i \lambda x_1^2/2\right] d\{x_1\}$

Out[∘]=

$$\frac{(-1)^{1/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-i \lambda x_1^2/2\right] d\{x_1\}$

Out[∘]=

$$-\frac{(-1)^{3/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[\frac{i}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\}\right] d\{x_1, x_2\}$

Out[∘]=

$$\frac{\mathbb{E}[0]}{2\sqrt{b^2 - a\,c}\;\pi}$$

In[∘]:= $\int \mathbb{E}\left[-\lambda x_1^2/2\right] d\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-x_1^2/2 + \xi\, x_1\right] d\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{\xi^2}{2}\right]}{\sqrt{2\pi}}$$

In[∘]:= $\int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\}\right] d\{x_1, x_2\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2 - 2\,b\,\xi_1\,\xi_2 + a\,\xi_2^2}{2\left(-b^2 + a\,c\right)}\right]}{2\sqrt{-b^2 + a\,c}\;\pi}$$

In[∘]:= $I1 = \int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\}\right] d\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}\left[-\frac{\left(-b^2 + a\,c\right)x_2^2}{2\,a} + \frac{\xi_1^2}{2\,a} + \frac{x_2\left(-b\,\xi_1 + a\,\xi_2\right)}{a}\right]}{\sqrt{a}\,\sqrt{2\pi}}$$

In[•]:= $\int \mathrm{I1}\, d\{x_2\}$

Out[•]=

$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2-2\,b\,\xi_1\,\xi_2+a\,\xi_2^2}{2\,(-b^2+a\,c)}\right]}{2\,\sqrt{-b^2+a\,c}\;\pi}$$

In[•]:= $\int \mathbb{E}\left[-\frac{1}{2}\,\{y_1, y_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{y_1, y_2\} + \{\eta_1, \eta_2\}.\{y_1, y_2\}\right] d\{y_1, y_2\}$

Out[•]=

$$\frac{\mathbb{E}\left[\frac{c\,\eta_1^2-2\,b\,\eta_1\,\eta_2+a\,\eta_2^2}{2\,(-b^2+a\,c)}\right]}{2\,\sqrt{-b^2+a\,c}\;\pi}$$

In[•]:= $\mathrm{I1} = \int \mathbb{E}\left[-\frac{1}{2}\,\{y_1, y_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{y_1, y_2\} + \{\eta_1, \eta_2\}.\{y_1, y_2\}\right] d\{y_1\}$

Out[•]=

$$\frac{\mathbb{E}\left[\frac{b^2\,y_2^2-a\,c\,y_2^2-2\,b\,y_2\,\eta_1+\eta_1^2+2\,a\,y_2\,\eta_2}{2\,a}\right]}{\sqrt{a}\;\sqrt{2\,\pi}}$$

In[•]:= $\int \mathrm{I1}\, d\{y_2\}$

Out[•]=

$$\frac{\mathbb{E}\left[\frac{a\,c\,\eta_1^2-2\,a\,b\,\eta_1\,\eta_2+a^2\,\eta_2^2}{2\,a\,(-b^2+a\,c)}\right]}{2\,\sqrt{-b^2+a\,c}\;\pi}$$

In[•]:= $\int \mathbb{E}\left[\xi\,x + \eta\,y + z\,(x - y) + x^2\right] d\{x, z\}$

Out[•]=

$$-\frac{i\,\mathbb{E}[y\,(y+\eta+\xi)]}{2\,\pi}$$

The ρ_1 Integrand

Adopted from pensieve://Projects/APAI/PerturbedGaussianIntegration.nb.

In[∘]:=
```
q[s_, i_, j_] := x_i ((1 - T^s) p_{i+1} + (T^s - 1) p_{j+1});
r_1[s_, i_, j_] :=
  s (-1 + 2 p_i x_i - 2 p_j x_i + (T^s - 1) p_i p_j x_i^2 + (1 - T^s) p_j^2 x_i^2 - 2 p_i p_j x_i x_j + 2 p_j^2 x_i x_j) / 2;
γ_1[φ_, k_] := ϵ φ (1 / 2 - x_k p_k);
ρ1i[s_, i_, j_] := T^(-s/2) 𝔼[-q[s, i, j] + ϵ r_1[s, i, j] + O[ϵ]^2];
ρ1i[φ_, k_] := T^(-φ/2) 𝔼[-x_k (p_k - p_{k+1}) + γ_1[φ, k] + O[ϵ]^2];
ρ1i[End, k_] := 𝔼[-x_k p_k + O[ϵ]^2];
ρ1i[K_] := Module[{Cs, φ, n, c, k, ℰ},
   {Cs, φ} = Rot[K]; n = Length[Cs];
   ℰ = ρ1i[End, 2 n + 1];
   Do[ℰ *= ρ1i @@ c, {c, Cs}];
   Do[ℰ *= ρ1i[φ⟦k⟧, k], {k, 2 n}];
   CF@ℰ
  ];
ρ1vs[K_] := Union @@ Table[{x_i, p_i}, {i, 2 Crossings[K] + 1}]
```

In[∘]:=
```
ρ1i[Knot[3, 1]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]=

$$T^2\,\mathbb{E}\Big[\Big(-((p_1-p_2)\,x_1)-(p_2-p_3)\,x_2-\frac{(-1+T)\,(p_3-p_6)\,x_2}{T}-(p_3-p_4)\,x_3+\frac{(-1+T)\,(p_2-p_5)\,x_4}{T}-(p_4-p_5)\,x_4-(p_5-p_6)\,x_5+\frac{(-1+T)\,(p_4-p_7)\,x_6}{T}-(p_6-p_7)\,x_6-p_7\,x_7\Big)+\Big(-\frac{1}{2}+p_4\,x_4+\frac{1}{2}\Big(1+2\,p_1\,x_4-2\,p_4\,x_4-2\,p_1^2\,x_1\,x_4+2\,p_1\,p_4\,x_1\,x_4-\Big(1-\frac{1}{T}\Big)\,p_1^2\,x_4^2-\Big(-1+\frac{1}{T}\Big)\,p_1\,p_4\,x_4^2\Big)+\frac{1}{2}\Big(1-2\,p_2\,x_2+2\,p_5\,x_2-\Big(-1+\frac{1}{T}\Big)\,p_2\,p_5\,x_2^2-\Big(1-\frac{1}{T}\Big)\,p_5^2\,x_2^2+2\,p_2\,p_5\,x_2\,x_5-2\,p_5^2\,x_2\,x_5\Big)+\frac{1}{2}\Big(1+2\,p_3\,x_6-2\,p_6\,x_6-2\,p_3^2\,x_3\,x_6+2\,p_3\,p_6\,x_3\,x_6-\Big(1-\frac{1}{T}\Big)\,p_3^2\,x_6^2-\Big(-1+\frac{1}{T}\Big)\,p_3\,p_6\,x_6^2\Big)\Big)\,\epsilon+O[\epsilon]^2\Big]$$

In[∘]:=
```
ρ1vs[Knot[3, 1]]
```

Out[∘]=

$\{p_1, p_2, p_3, p_4, p_5, p_6, p_7, x_1, x_2, x_3, x_4, x_5, x_6, x_7\}$

Integration of ϵ-Series

Using Picard Iteration!

In[∘]:=
```
Unprotect[Integrate];
Integrate::sing = "How dare you ask me to integrate a singular Gaussian!";
∫ω_. 𝔼[L_SeriesData] ⅆ(vs_List) := Module[{n, m, ϵ, L0, Q, Δ, G, Z, e, λ, a, b},
   ϵ = L〚1〛; m = L〚5〛;
   n = Length@vs; L0 = Normal@L /. ϵ → 0;
   Q = Table[(-∂_{vs〚a〛,vs〚b〛} L0) /. Thread[vs → 0] /. (p | x)__ → 0, {a, n}, {b, n}];
   If[(Δ = CF@Det[Q]) == 0, Message[Integrate::sing]; Return[]];
   Z = L + vs.Q.vs / 2; G = Inverse[Q];
   While[
     e = Normal@CF[(∂_λ Z) - 1/2 Sum_{a=1}^{n} Sum_{b=1}^{n} ((∂_{vs〚a〛,vs〚b〛} Z) + (∂_{vs〚a〛} Z) (∂_{vs〚b〛} Z) + O[ϵ]^m)];
     0 =!= e, Z -= ∫_0^λ e ⅆλ
   ];
   PowerExpand@Factor[ω (Δ (2 π)^n)^(-1/2)] 𝔼[CF[Z /. λ → 1 /. Thread[vs → 0]]]
  ];
Protect[Integrate];
```

In[∘]:=
```
∫𝔼[-x^2 / 2 + δ x^3 / 6 + O[δ]^20] ⅆ{x}
```

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{5\delta^2}{24}+\frac{5\delta^4}{16}+\frac{1105\delta^6}{1152}+\frac{565\delta^8}{128}+\frac{82825\delta^{10}}{3072}+\frac{19675\delta^{12}}{96}+\frac{1282031525\delta^{14}}{688128}+\frac{80727925\delta^{16}}{4096}+\frac{1683480621875\delta^{18}}{7077888}+O[\delta]^{20}\right]}{\sqrt{2\pi}}$$

In[∘]:=
```
∫𝔼[-ϕ^2 / 2 + ϵ ϕ^4 / 24 + O[ϵ]^16] ⅆ{ϕ}
```

Out[∘]=

$$\frac{1}{\sqrt{2\pi}}\mathbb{E}\left[\frac{\epsilon}{8}+\frac{\epsilon^2}{12}+\frac{11\epsilon^3}{96}+\frac{17\epsilon^4}{72}+\frac{619\epsilon^5}{960}+\frac{709\epsilon^6}{324}+\frac{858437\epsilon^7}{96768}+\frac{54193\epsilon^8}{1296}+\frac{18639247\epsilon^9}{82944}+\frac{2197187\epsilon^{10}}{1620}+\frac{33152545703\epsilon^{11}}{3649536}+\frac{1169890097\epsilon^{12}}{17496}+\frac{41657327595361\epsilon^{13}}{77635584}+\frac{31722037141\epsilon^{14}}{6804}+\frac{6944870083473751\epsilon^{15}}{159252480}+O[\epsilon]^{16}\right]$$

In[∘]:=
```
∫𝔼[p x + ϵ a p^2 x + O[ϵ]^5] ⅆ{p, x}
```

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[\frac{15a^2\epsilon^2}{2}+405a^4\epsilon^4+O[\epsilon]^5\right]}{2\pi}$$

In[∘]:= $\int \mathbb{E}\left[p\,x + \epsilon\,a\,p^2\,x + \epsilon\,b\,p\,x^2 + O[\epsilon]^5\right] \,d\{p, x\}$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[\frac{15}{2}\,(a+b)^2\,\epsilon^2 + 405\,(a+b)^4\,\epsilon^4 + O[\epsilon]^5\right]}{2\,\pi}$$

$\int \mathbb{E}\left[p\,x + p^2\,x + O[\epsilon]^5\right] \,d\{p, x\}$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[\frac{15\,\epsilon^2}{2} + 405\,\epsilon^4 + O[\epsilon]^5\right]}{2\,\pi}$$

In[∘]:= $\int \mathbb{E}\left[x_1\,p_1 + \epsilon\,x_1^7\,p_1^7 + O[\epsilon]^2\right] \,d\{x_1, p_1\}$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-5040\,\epsilon + O[\epsilon]^2\right]}{2\,\pi}$$

In[∘]:= $\int \mathbb{E}\left[x_1\,p_2 + \epsilon\,x_2^7\,p_1^7 + O[\epsilon]^2\right] \,d\{x_1, p_2\}$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[p_1^7\,x_2^7\,\epsilon + O[\epsilon]^2\right]}{2\,\pi}$$

In[∘]:= $\int \mathbb{E}\left[x_1\,p_2 + 3\,x_2\,p_1 + \epsilon\,p_2^5\,x_1^5 + O[\epsilon]^2\right] \,d\{x_1, x_2, p_1, p_2\}$

Out[∘]=

$$\frac{\mathbb{E}\left[-120\,\epsilon + O[\epsilon]^2\right]}{12\,\pi^2}$$

In[∘]:= $\int \mathbb{E}\left[x_1\,p_2 + x_2\,p_3 + x_3\,p_1 + \epsilon\,x_1^5\,p_2^5 + O[\epsilon]^2\right] \,d\{x_1, x_2, x_3, p_1, p_2, p_3\}$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-120\,\epsilon + O[\epsilon]^2\right]}{8\,\pi^3}$$

In[∘]:= MatrixForm@Table[
$\int \mathbb{E}[x_1\,p_2 + x_2\,p_3 + x_3\,p_1 + \xi_i\,x_i + \pi_j\,p_j + O[\epsilon]] \,d\{x_1, x_2, x_3, p_1, p_2, p_3\}$,
{i, 3}, {j, 3}]

Out[∘]//MatrixForm=

$$\begin{pmatrix} -\frac{i\,\mathbb{E}[O[\epsilon]^1]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[-\pi_2\,\xi_1 + O[\epsilon]^1]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[O[\epsilon]^1]}{8\,\pi^3} \\ -\frac{i\,\mathbb{E}[O[\epsilon]^1]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[O[\epsilon]^1]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[-\pi_3\,\xi_2 + O[\epsilon]^1]}{8\,\pi^3} \\ -\frac{i\,\mathbb{E}[-\pi_1\,\xi_3 + O[\epsilon]^1]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[O[\epsilon]^1]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[O[\epsilon]^1]}{8\,\pi^3} \end{pmatrix}$$

In[∘]:=
```
K = Knot[5, 2];
∫ρ1i[K] ⅆ(ρ1vs@K)
```

Out[∘]=

$$-\frac{i\, T^7\, \mathbb{E}\left[\frac{(-1+T)^2\,(5-4T+5T^2)\,\epsilon}{(2-3T+2T^2)^2}+O[\epsilon]^2\right]}{2048\,\pi^{11}\,(2-3T+2T^2)}$$

In[∘]:=
```
K = Knot[8, 19];
∫ρ1i@K ⅆ(ρ1vs@K)
```

Out[∘]=

$$-\frac{i\, \mathbb{E}\left[-\frac{(-1+T)^2\,(1+T^4)\,(3+4T^3+3T^6)\,\epsilon}{(1-T+T^2)^2\,(1-T^2+T^4)^2}+O[\epsilon]^2\right]}{131072\,\pi^{17}\,T^5\,(1-T+T^3-T^5+T^6)}$$

Invariance Under Reidemeister 3b

In[∘]:=
```
lhsi = ρ1i[1, i, j] ρ1i[1, i + 1, k] ρ1i[1, j + 1, k + 1]
   ρ1i[0, i] ρ1i[0, j] ρ1i[0, k] ρ1i[0, i + 1] ρ1i[0, j + 1] ρ1i[0, k + 1];
lhs = ∫lhsi ⅆ{x_{i+1}, x_{j+1}, x_{k+1}, p_{i+1}, p_{j+1}, p_{k+1}}
```

Out[∘]=

$$-\frac{1}{8\,\pi^3\,T^{3/2}}\, i\, \mathbb{E}\Big[\big(-p_i x_i + T^2 p_{2+i} x_i - (-1+T)\,T\,p_{2+j} x_i + (1-T)\,p_{2+k} x_i - p_j x_j + T\,p_{2+j} x_j + (1-T)\,p_{2+k} x_j - p_k x_k + p_{2+k} x_k\big) + \Big(-\frac{1}{2} + p_i x_i + T^2 p_{2+i} x_i - p_j x_i + (-2+T)\,T\,p_k x_i - (-1+T)\,T\,p_{2+k} x_i + \frac{1}{2}(-1+T)\,p_i p_j x_i^2 + \frac{1}{2}(1-T)\,p_j^2 x_i^2 + \frac{1}{2}(-1+T)\,T^3 p_{2+i} p_k x_i^2 - \frac{1}{2}(-1+T)\,T^2 p_k^2 x_i^2 - \frac{1}{2}(-1+T)^2\,T\,(1+T)\,p_{2+j} p_{2+k} x_i^2 - \frac{1}{2}(-1+T)^2\,T^2 p_k p_{2+k} x_i^2 + \frac{1}{2}(-1+T)^2\,T\,(1+T)\,p_{2+k}^2 x_i^2 + T\,p_{2+k} x_j - p_i p_j x_i x_j + p_j^2 x_i x_j + (-1+T)\,T\,p_{2+j} p_{2+k} x_i x_j - (-1+T)\,T\,p_{2+k}^2 x_i x_j + \frac{1}{2}(-1+T)\,T\,p_{2+j} p_{2+k} x_j^2 - \frac{1}{2}(-1+T)\,T\,p_{2+k}^2 x_j^2 - p_k x_k - p_{2+k} x_k - T^2 p_{2+i} p_k x_i x_k + T\,p_k^2 x_i x_k + (-1+T)\,T\,p_{2+j} p_{2+k} x_i x_k + (-1+T)\,T\,p_k p_{2+k} x_i x_k - (-1+T)\,T\,p_{2+k}^2 x_i x_k - T\,p_{2+j} p_{2+k} x_j x_k + T\,p_{2+k}^2 x_j x_k\Big)\,\epsilon + O[\epsilon]^2\Big]$$

In[∘]:=
```
rhsi = ρ1i[1, j, k] ρ1i[1, i, k + 1] ρ1i[1, i + 1, j + 1]
   ρ1i[0, i] ρ1i[0, j] ρ1i[0, k] ρ1i[0, i + 1] ρ1i[0, j + 1] ρ1i[0, k + 1];
rhs = ∫lhsi ⅆ{x_{i+1}, x_{j+1}, x_{k+1}, p_{i+1}, p_{j+1}, p_{k+1}};
```

In[∘]:= **lhs == rhs**

Out[∘]=

True

Invariance Under Reidemeister 2b

In[∘]:= **lhsi = ρ1i[1, i, j] ρ1i[-1, i + 1, j + 1] ρ1i[0, i] ρ1i[0, j] ρ1i[0, i + 1] ρ1i[0, j + 1];**

lhs = ∫lhsi ⅆ{x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}

Out[∘]=

$$\frac{1}{4\pi^2}\mathbb{E}\Big[(-p_i x_i + p_{2+i} x_i - p_j x_j + p_{2+j} x_j) + \Big(p_i x_i - p_j x_i - T p_{2+j} x_i + \frac{1}{2}(-1+T) p_i p_j x_i^2 + \frac{1}{2}(1-T) p_j^2 x_i^2 + \frac{1}{2}(1-T) p_{2+i} p_{2+j} x_i^2 + \frac{1}{2}(-1+T) p_{2+j}^2 x_i^2 + p_{2+j} x_j - p_i p_j x_i x_j + p_j^2 x_i x_j + p_{2+i} p_{2+j} x_i x_j - p_{2+j}^2 x_i x_j\Big) \in + O[\in]^2\Big]$$

In[∘]:= $$\mathbf{CF}\Big[\Big(p_i x_i - p_j x_i - T p_{2+j} x_i + \frac{1}{2}(-1+T) p_i p_j x_i^2 + \frac{1}{2}(1-T) p_j^2 x_i^2 + \frac{1}{2}(1-T) p_{2+i} p_{2+j} x_i^2 + \frac{1}{2}(-1+T) p_{2+j}^2 x_i^2 + p_{2+j} x_j - p_i p_j x_i x_j + p_j^2 x_i x_j + p_{2+i} p_{2+j} x_i x_j - p_{2+j}^2 x_i x_j\Big) \;/.\; p_{k\_+2} \to p_k\Big]$$

Out[∘]=

$$p_i x_i + (-1-T) p_j x_i + p_j x_j$$

In[∘]:= **lhsi = ρ1i[1, i, j] ρ1i[-1, i + 1, j + 1] ρ1i[0, i] ρ1i[0, j] ρ1i[0, i + 1] ρ1i[0, j + 1];**

lhs = ∫𝔼[π_i p_i + π_j p_j] lhsi ⅆ{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}

Out[∘]=

$$\frac{\mathbb{E}\left[(p_{2+i}\pi_i + p_{2+j}\pi_j) + O[\in]^2\right]}{16\pi^4}$$

In[∘]:= **lhs =**

∫𝔼[π_i p_i + π_j p_j + ℒ[1, i, j] + ℒ[-1, i + 1, j + 1] + O[∈]^2] ⅆ{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}

Out[∘]=

$$\frac{\mathbb{E}\left[(p_{2+i}\pi_i + p_{2+j}\pi_j) + O[\in]^2\right]}{16\pi^4}$$

In[∘]:= **rhs = ∫𝔼[π_i p_i + π_j p_j + ℒ[0, i] + ℒ[0, i + 1] + ℒ[0, j] + ℒ[0, j + 1] + O[∈]^2]**

ⅆ{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}

Out[∘]=

$$\frac{\mathbb{E}\left[(p_{2+i}\pi_i + p_{2+j}\pi_j) + O[\in]^2\right]}{16\pi^4}$$

In[∘]:= **lhs == rhs**

Out[∘]=

True

Invariance Under R2c

In[∘]:= $\text{lhs} = \int \mathbb{E}\left[\pi_i\, p_i + \pi_j\, p_j + \mathcal{L}[-1, i, j+1] + \mathcal{L}[1, i+1, j] + \gamma_1[-1, j+1] + O[\epsilon]^2\right] d\{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}\}$

Out[∘]=

$$\frac{\mathbb{E}\left[(p_{2+i}\,\pi_i + p_{2+j}\,\pi_j) + \frac{\epsilon}{2} + O[\epsilon]^2\right]}{16\,\pi^4}$$

In[∘]:= $\text{rhs} = \int \mathbb{E}\left[\pi_i\, p_i + \pi_j\, p_j + \mathcal{L}[0, i] + \mathcal{L}[0, i+1] + \mathcal{L}[0, j] + \mathcal{L}[-1, j+1] + O[\epsilon]^2\right] d\{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}\}$

Out[∘]=

$$\frac{\mathbb{E}\left[(p_{2+i}\,\pi_i + p_{2+j}\,\pi_j) + \frac{\epsilon}{2} + O[\epsilon]^2\right]}{16\,\pi^4}$$

In[∘]:= **lhs == rhs**

Out[∘]=

True

Invariance Under R1l

In[∘]:= $\text{lhs} = \int \mathbb{E}\left[\pi_i\, p_i + \mathcal{L}[1, i+2, i] + \mathcal{L}[1, i+1] + O[\epsilon]^2\right] d\{x_i, x_{i+1}, x_{i+2}, p_i, p_{i+1}, p_{i+2}\}$

$$\gg \begin{pmatrix} 0 & 0 & 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 0 & -1+T & 1 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ -1 & 1 & -1+T & 0 & 0 & 0 \\ 0 & -1 & 1 & 0 & 0 & 0 \end{pmatrix}$$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[p_{3+i}\,\pi_i + O[\epsilon]^2\right]}{8\,\pi^3\,T}$$

In[ ]:= **rhs** = $\int \mathbb{E}\left[\pi_i\, p_i + \mathcal{L}[0, i] + \mathcal{L}[0, i+1] + \mathcal{L}[0, i+2] + O[\epsilon]^2\right] d\{x_i, x_{i+1}, x_{i+2}, p_i, p_{i+1}, p_{i+2}\}$

» $\begin{pmatrix} 0 & 0 & 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ -1 & 1 & 0 & 0 & 0 & 0 \\ 0 & -1 & 1 & 0 & 0 & 0 \end{pmatrix}$

Out[ ]=

$$-\frac{i\, \mathbb{E}\left[p_{3+i}\, \pi_i + O[\epsilon]^2\right]}{8\, \pi^3}$$

In[ ]:= **lhs == rhs**

Out[ ]=

$$-\frac{i\, \mathbb{E}\left[p_{3+i}\, \pi_i + O[\epsilon]^2\right]}{8\, \pi^3\, T} == -\frac{i\, \mathbb{E}\left[p_{3+i}\, \pi_i + O[\epsilon]^2\right]}{8\, \pi^3}$$